*=============**PWM_WithOut_PM_Sanity**=============================

A good sanity test as to whether a balanced PWM produces no phase modulation, is to linearly bandpass filter the PWM signal, and then clip it.



The Q of the bandpass filter does need to be set high enough to remove the modulating second harmonic. The output of the bandpass should still show some amplitude modulation.



Clipping the signal after the bandpass should only remove AM. PM should mainly be unaffected.

```
PWM_Without_PM_BP
*
*               VTRI          VPWM          VBP  _______________
*                                                |              |      ^  OUT
*  VT      _____|     _____|     _____|           |  _ _ _        |     / \
*        _|_    |   _|_    |   _|_    |       R1  |  / \/ \/ \     |    /   \
*       /   \   |  /   \   |  /   \   |_/\  /\  /\__|_| () () |_    |   _|___
*      //   \   |_//   \   |_//   \      \/  \/   |    L1    _|_  |   /    \
*      \    \// \      \//   \    \//             _|_       ///  |_//    \
*       \___/    \____/       \___/              ___ C1              \   \_//
*         |    B1   |     B2   |                  |                   \____/
*        _|_       _|_        _|_                _|_                   _|_
*        ///       ///        ///                ///                   ///
*
VP        VP        0     DC      3.141592653589793
VT        VT        0     PWL     ( 0 0 1 1 )
VIN       VIN       0     DC      1 AC 1
B1        VTRI      0     V =     acos(cos(2*v(VP)*20*v(VT)))/v(VP)
B2        VPWM      0     V =     2.0*u(1-v(VTRI)*2  + .9*cos(1*v(VP)*2*v(VT)) )-1
B3        VREF      0     V =     1.8*u(1-v(VTRI)*2  )-.9
B4        VOUT      0     V =     1.4*u( v(VBP)*2    ) -.7
R1        VPWM      VBP   34000
L1        VBP       0     6.333   IC= 0
C1        VBP       0     10u     IC= .849
*=========Run_Simulation============================================
.tran     .1m       1     0      .1m   UIC
.control
run
set       pensize = 2
plot      vpwm vout
plot      vref vpwm vout xlimit 0 .05
plot      vref vpwm vout xlimit .5 .55
plot      vref vpwm vout xlimit .95 1
.endc
.end
*==================================================================
```

The amplitude of the reference square wave and the output square wave are made slightly different on the plots below to display the fact that they are pretty much the same curve.





The PWM signal is also included for comparison.

Looks like the unmodulated square wave can be recovered from this type of pulse width modulation by just using a bandpass filter and an inverter.

===========Full_Netlist_For_Copy_Paste=======================

```
PWM_Without_PM_BP
*
*              VTRI          VPWM          VBP ________________
*                   _____         _____         _____      |                |      ^  OUT
* VT      _|_   |     _|_   |     _|_   |       R1     |   _   _   _ |      / \
*        /_  \  |    /_  \  |    /_  \  |_/\  /\  /\__|_| () () |_     _|__
*       //\  \ |_//\  \ |_//\  \    \/  \/    |           L1  _|_ |   /_   \
*       \   \// \   \//  \   \//               _|_              /// |_//\  \_//
*        \___/      \___/     \___/               ___  C1               \    /
*          |    B1     |    B2    |                |                      \___/
*         _|_         _|_        _|_              _|_                     _|_
*         ///         ///        ///              ///                     ///
*
VP        VP        0     DC     3.141592653589793
VT        VT        0     PWL    ( 0 0 1 1 )
VIN       VIN       0     DC     1 AC 1
B1        VTRI      0     V =    acos(cos(2*v(VP)*20*v(VT)))/v(VP)
B2        VPWM      0     V =    2*u(1-v(VTRI)*2  + .9*cos(1*v(VP)*2*v(VT)) )-1
B3        VREF      0     V =    1.8*u(1-v(VTRI)*2    )-.9
B4        VOUT      0     V =    1.4*u( v(VBP)*2    )-.7
R1        VPWM       VBP  34000
L1        VBP       0     6.333  IC= 0
C1        VBP       0     10u    IC= .849
.tran     .1m       1  0     .1m   UIC
.control
run
set       pensize = 2
plot      vpwm vout
plot      vout vbp
plot      vref vpwm vout xlimit 0 .05
plot      vref vpwm vout xlimit .5 .55
plot      vref vpwm vout xlimit .95 1
.endc
.end
```

2.12.10_3.08PM

dsauersanjose@aol.com
Don Sauer
http://www.idea2ic.com/